

```

theory Warmup-Problem-D
imports
  Complex-Main
  Common.Future-Library
  HOL-Analysis.Analysis
begin

```

0.1 Warmup problem D

There is a set of n points on a plane with the property that, in each triplet of points, there's a pair with distance at most 1. Prove that the set can be covered with two circles of radius 1.

There's nothing special about the case of points on a plane, the theorem can be proved without additional difficulties for any metric space:

```

theorem warmup4-generic:
  fixes  $S :: 'a::metric-space\ set$ 
  assumes  $finite\ S$ 
  assumes  $property: \bigwedge T. T \subseteq S \wedge card\ T = 3 \implies \exists p \in T. \exists q \in T. p \neq q \wedge dist\ p\ q \leq 1$ 
  obtains  $O_1\ O_2$  where  $S \subseteq cball\ O_1\ 1 \cup cball\ O_2\ 1$ 
proof
  let  $?pairs = S \times S$ 
  let  $?dist = \lambda(a, b). dist\ a\ b$ 
  define  $widest-pair$  where  $widest-pair = arg-max-on\ ?dist\ ?pairs$ 
  let  $?O_1 = (fst\ widest-pair)$ 
  let  $?O_2 = (snd\ widest-pair)$ 
  show  $S \subseteq cball\ ?O_1\ 1 \cup cball\ ?O_2\ 1$ 
proof
  fix  $x$ 
  assume  $x \in S$ 

  from  $\langle finite\ S \rangle$  and  $\langle x \in S \rangle$ 
  have  $finite\ ?pairs$  and  $?pairs \neq \{\}$  by  $auto$ 
  hence  $OinS: widest-pair \in ?pairs$ 
  unfolding  $widest-pair-def$  by  $(simp\ add: arg-max-if-finite)$ 

  have  $\forall (P, Q) \in ?pairs. dist\ ?O_1\ ?O_2 \geq dist\ P\ Q$ 
  unfolding  $widest-pair-def$ 
  using  $\langle finite\ ?pairs \rangle$  and  $\langle ?pairs \neq \{\} \rangle$ 
  by  $(metis\ (mono-tags,\ lifting)\ arg-max-greatest\ prod.case-eq-if)$ 
  hence  $greatest: dist\ P\ Q \leq dist\ ?O_1\ ?O_2$  if  $P \in S$  and  $Q \in S$  for  $P\ Q$ 
  using  $that$  by  $blast$ 

  let  $?T = \{?O_1, ?O_2, x\}$ 
  have  $TinS: ?T \subseteq S$  using  $OinS$  and  $\langle x \in S \rangle$  by  $auto$ 

  have  $card\ ?T = 3$  if  $?O_1 \neq ?O_2$  and  $x \notin \{?O_1, ?O_2\}$  using  $that$  by  $auto$ 
  then consider
     $(primary)\ card\ ?T = 3 \mid$ 
     $(limit)\ x \in \{?O_1, ?O_2\} \mid$ 
     $(degenerate)\ ?O_1 = ?O_2$  by  $blast$ 
  thus  $x \in cball\ ?O_1\ 1 \cup cball\ ?O_2\ 1$ 
proof  $cases$ 
  case  $primary$ 
  obtain  $p$  and  $q$  where  $p \neq q$  and  $dist\ p\ q \leq 1$  and  $p \in ?T$  and  $q \in ?T$ 
  using  $property\ [of\ ?T]$  and  $\langle card\ ?T = 3 \rangle\ TinS$ 
  by  $auto$ 
  then have
     $dist\ ?O_1\ ?O_2 \leq 1 \vee dist\ ?O_1\ x \leq 1 \vee dist\ ?O_2\ x \leq 1$ 
    by  $(metis\ dist-commute\ insertE\ singletonD)$ 
  thus  $x \in cball\ ?O_1\ 1 \cup cball\ ?O_2\ 1$ 
  using  $greatest$  and  $TinS$ 

```

```

      by fastforce
next
  case limit
  then have  $\text{dist } x \text{ ?}O_1 = 0 \vee \text{dist } x \text{ ?}O_2 = 0$  by auto
  thus ?thesis by auto
next
  case degenerate
  with greatest and TinS have  $\text{dist ?}O_1 x = 0$  by auto
  thus ?thesis by auto
qed
qed
qed

```

Let's make sure that the particular case of points on a plane also works out:

```

corollary warmup4:
  fixes  $S :: (\text{real}^2)$  set
  assumes finite S
  assumes property:  $\bigwedge T. T \subseteq S \wedge \text{card } T = 3 \implies \exists p \in T. \exists q \in T. p \neq q \wedge \text{dist } p \ q \leq 1$ 
  obtains  $O_1 \ O_2$  where  $S \subseteq \text{cball } O_1 \ 1 \cup \text{cball } O_2 \ 1$ 
  using warmup4-generic and assms by auto
end

```